

What's on my Mainframe?

A macro that gives you a solid overview of your data on the mainframe.

Jesper Michelsen, Group Risk Management, Nykredit A/S

ABSTRACT

In connection with the consolidation of our SAS® platform at Nykredit the data stored on the Nykredit z/OS SAS® installation had to be migrated (copied) to the new x64 Windows SAS platform storage.

However, getting an overview of these data on the z/OS mainframe can be difficult, and a series of questions arise during the process. For example: Who is responsible? How many bytes? How many rows and columns? When were the data created? And so on.

With extensive use of filename FTP and looping, and extracting metadata, it is possible to get an overview of the data on the host presented in an Excel sheet.

INTRODUCTION

During 2013 we at Nykredit has consolidated our SAS hardware platform from 250+ local installations and a z/OS SAS and data server to an almost solely Windows Server x64 based platform.

This consolidation to Microsoft Windows and the migration of data from z/OS to a new data storage has reduced the 'analytic' workload on our z/OS with more than 80%. The reduction in the consuming of MIPS paid the project. As a side effect many of the jobs runs 50-70% faster on the new platform and it has become easier to automate and schedule jobs.

About 1.700 data catalogues at the z/OS data server had to be investigated and migrated to the new data storage.

We discovered that most users lacked the necessary skills to work in the 3270-environment, and it was difficult for them to get a total overview of all of their data catalogues in this environment.

The users could allocate the catalogues as a libref one by one in SAS® and investigate the content. But that is time consuming – and boring to do by hand.

A little creative thinking and the SAS® FILENAME statement, FTP Access method, and the included FTP Options provided us with the solution to this task.

With the SAS® FILENAME statement, using the FTP Access Method you can 'remote' control FTP commands sent to your host – in our case the z/OS mainframe – through SAS®.

You don't have to provide the FTP commands with a fully qualified "catalogue -path" – it is possible to construct a "search-string" including an asterisk '*'

And you can even read the information from the "FTP filename" into a dataset.

But - if your host or a firewall isn't opened for FTP you have to speak to your systems manager

STEP-BY-STEP

The first step is to construct the right FILENAME statement

- it is possible to construct a search string for the LSFIL= including an asterisk '*'

```
filename ftplist ftp ' ls cd='..' lsfile="&_ds..*" user="&_usr"  
host="&_host" pass=&_password;
```

The LS command is issued and that the LSFIL command controls for which part of the data catalogues the list should be made. This gives you the desired list of data catalogues on the z/OS mainframe.

The _ds macro variable contains either the fully qualified "DSN path" or a part of it.

The second step is to read the list into a SAS® dataset

- you probably have to know a little about how the list will look like

```
data DSNdata01;
    length DSNx $ 128;
    infile ftplist missover pad;
    input DSNx $128.;
Run;
```

Special characters like 'ÆØÅæøå' are important as they have a special meaning on a z/OS – you may need to translate in some way to get a reasonable result.

The third step is to loop through the dataset allocating the z/OS catalogues one by one

- we used some of the functions available to %sysfunc – open, fetchobs, getvarc, close
- in each loop we allocated a new libname and retrieved the desired information through the dictionary tables
- how many loops is decided by the number of records in the above dataset

We chose to extract the following information for each member in the catalogue

- name, memtype, crdate, modate, nob, nvar, obslen, filesize and encoding
- for administrative use we also added some 'empty columns' e.g. (new)destination, status, who is responsible

The information is then appended to a master dataset which eventually is exported to an Excel workbook using a SAS® LIBNAME statement for Excel.

The whole sequence was built in to a macro to make it easier for users to retrieve the informations they needed

- actually we made two macros
- one which just gives a list of the catalogues on the z/OS from a certain search-point
- one which extracts the above mentioned information from a certain search-point

THE MACRO

The macro needs four parameters

- `_host`; which is the name of the mainframe server
- `_ds`; which is the "search-string"; e.g. KAF.SAS.SASDATA.IAS*
- `_complete`; indicates whether the search-string is fully qualified or not
- `_excelpath`; the path to where the resulting excel workbooks should be located. The macro is naming the workbook file by itself.

```
1 %macro listDSNdata2Excel(_host, _ds, _complete, _excelpath);
2 options msglevel=N;
3 %local _i _dsid _disp _excelpath _rc _rclib _dsn _host _complete _dsx _file;
4 %global _usr;
5 %readpwkrypt(&_host);
6 %if %upcase(&_complete) = C %then
7 %do;
8     filename ftplist ftp ' ' ls cd='..' lsfile="&_ds" user="&_usr" host="&_host"
9         pass=&_password;
10 %end;
11 %else
12 %if %upcase(&_complete) = U %then
13 %do;
```

```

14     filename ftplist ftp ' ' ls cd='../' lsfile="&_ds..*" user="&_usr"
15         host="&_host" pass=&_password;
16     %end;
17 %if %eval(&sysfilrc=0) %then
18     %do;
19         proc datasets lib=work kill;
20         run;quit;

21     data DSNdata01;
22     length DSNx $ 128;
23     infile ftplist missover pad;
24     input DSNx $128.;
25     run;

26     data DSNdata01(drop=DSNx);
27     set DSNdata01;
28     DSN=translate(DSNx,'Ø','9D'x);
29     run;

30     proc sort data=dsndata01;
31     by dsn;
32     run;
33 %end;
34 %else
35 %do;
36     %PUT WARNING: Cannot find any DSN with these qualifiers: %upcase(&_ds.);
37     %put WARNING: Correct and try again;
38 %end;
39     filename ftplist clear;

/* Step 2: Allocate DSN and examine content */
40 %if %sysfunc(exist(work.DSNdata01))=1 %then
41     %do;
42         %son(&_host);
43         %nobs(work.DSNdata01);
44         %let _dsid=%sysfunc(open(DSNdata01));
45         %put _dsid: &_dsid;
46         %do _i=1 %to &_nobs;
47             %let _rc=%sysfunc(fetchobs(&_dsid,&_i));
48             %let _DSN=%sysfunc(getvarc(&_dsid,1));
49             libname HOST "&_DSN" server=SASNK02 access=readonly;
50             %let _rclib=%sysfunc(libref(HOST));

51             %if &_rclib=0 %then
52                 %do;
53                     proc sql;
54                         create table _dsnmembers&_i. as
55                         select resolve("&_DSN") as DSNHOST length=128 format=$128.
56                             ,memname as name
57                             ,memtype as type
58                             ,crdate
59                             ,modate
60                             ,nobs
61                             ,nvar
62                             ,obslen
63                             ,filesize
64                             ,encoding
65                         from dictionary.tables
66                         where libname='HOST';
67                     quit;

68             %if &sqlobs=0 %then
69                 %do;

```

```

70         data _dsnmembers&i.;
71         length DSNHOST $ 128
72         navn $ 32
73         type $ 8
74         crdate
75         modate
76         nob
77         nvar
78         obslen
79         filesize 8
80         encoding $ 256;
81         DSNHOST=resolve("&_DSN");
82         navn='DSN is empty';
83         type='N/A';
84         crdate=0;
85         modate=0;
86         nob=0;
87         nvar=0;
89         obslen=0;
90         filesize=0;
91         encoding='N/A';
92         run;
93     %end;
94 %end;
95 %else
96 %do;
97     data _dsnmembers&i.;
98     length DSNHOST $ 128
99     navn $ 32
100    type $ 8
101    crdate
102    modate
103    nob
104    nvar
105    obslen
106    filesize 8
107    encoding $ 256;
108
109    DSNHOST=resolve("&_DSN");
110    navn='N/A';
111    type='N/A';
112    crdate=0;
113    modate=0;
114    nob=0;
115    nvar=0;
116    obslen=0;
117    filesize=0;
118    encoding='DSN in use';
119    run;
120 %end;
121     proc append base=dsnmembers01
122             data=_dsnmembers&i force;
123     run;
124     libname HOST clear;
125 %end;
126 %let _rc=%sysfunc(close(&_dsid));
127 %put RC fra 'CLOSE': &_rc;
128
129 proc sql;
130 create table dsn as
131 select ' ' as plan label='Planned? Y/N'
        ,DSN label='DSN at Host'
        , ' ' as Migrate label='Migrate Y/N?'

```

```

132     ,'' as Method label='Method: Manuel/Batch'
133     ,'' as Destination label='Destination'
134     ,'' as Info_date label='Info on migration to (date)'
135     ,'' as Info_init label='Info om migration mail to'
136     ,'' as Info_OK label='OK (date)'
137     ,'' as DSN_Read_Only_date label='DSN R/O (date)'
138     ,'' as Control_date 'Migration controlled (date)'
139     ,'' as NO_Read_date 'DSN no-read/no-write (date)'
140 from dsndata01;

141 create table datasets as
142 select '' as plan label='Planned? Y/N'
143     ,'' as ansv label='Responsebility for migration of DSN'
144     ,DSNHOST label='DSN at Host'
145     ,navn label='Member Name in DSN'
146     ,type label='Member-type'
147     ,crdate label='Create date'
148     ,modate label='Modified date'
149     ,nobs label='No of rows'
150     ,nvar label='No of columns'
151     ,obslen label='No of bytes per row'
152     ,(filesize/1024) as filemb label='Filsize in Kb'
152     ,encoding label='Encoding'
154     ,'' as Migrate label='Migrate Y/N?'
155     ,'' as Method label='Method: Manuel/Batch'
156     ,'' as Destination label='Destination'
157     ,'' as Info_date label='Info on migration mailed (date)'
158     ,'' as Info_init label='Info on migration mailed to'
159     ,'' as Info_OK label='OK (date)'
160     ,'' as DSN_Read_Only_date label='DSN R/O (date)'
161     ,'' as Control_date 'Migration controlled (date)'
162     ,'' as NO_Read_date 'DSN no-read/no-write (date)'
163 from dsnmembers01;
164 quit;

165 %let dato=%sysfunc(today(),date9.);
166 %let time=%sysfunc(time(),hhmm8.);
167 %let _file=&_excelpath.\DSNdata %upcase(&_ds.) pr. &dato..xlsx;
168 %put NOTE: The EXCEL workbook is filed - &_file.;

169 libname myxlbook "&_file.";

170 %if %sysfunc(fileexist(&_file)) = 1 %then
171     %do;
172         proc datasets lib=myxlbook nolist;
173             delete dsnmembers02 dsndata02;
174             %_run(wq);
175     %end;

176 data myxlbook.DSN(dblabel=yes);
177 set dsn;
178 %_run(w);

179 data myxlbook.Datasets(dblabel=yes);
180 set datasets;
181 %_run(w);

182 libname myxlbook clear;

183 %end;
184 %soff(&_host);
185 %mend listDSNdata2Excel;

```

The readPWkrypt in-house Nykredit macro in line 5 reads a encrypted password stored in SASUSER. Initially the password was encrypted by the in-house Nykredit macro PWkrypt.

Line 6-16 is the FILENAME statements. Only one of them is executed, which depends on the value of the macro variable _complete. If _complete has the value U (for uncompleted)the macro variable _ds is extended with “.*” to indicate that the “search-string” is not fully qualified.

If the filename is a success (line 17) the sequence in line 21-32 is executed. If not a WARNING appears in the log. The rest of the macro fails to execute.

If the sequence in line 21-32 creates the dataset ‘DSNdata01’ then a signon to the host (the mainframe server) is executed. This is done through aa in-house Nykredit macro called ‘son’. After this the in-house Nykredit macro nobis retrieves the number of records from the dataset DSNdata01 and put the value into a macrovariable called _nobs.

The looping through the dataset is then initialized and for each loop the name of the mainframe DSN is retrieved and used for allocation of a libname called HOST.

The information needed is then selected from dictionary.tables where the libame is equal to ‘HOST’. If nothing ca be selected, a NULL record is constructed instead. If the DSN is in use or for some other reason not available a ‘DSN in use’-record is constructed.

The records are appended to a master dataset in line 120-122 and the dataset ‘DSNdata01’ is closed. After this the extra administrative empty columns is created (line 127-164).

Finally the datasets are exported to Excel (line 169-182) and the connection to the host is disconnected using an in-house Nykredit macro called ‘soff ‘.

CONCLUSION

SAS® FILENAME Statement with the FTP ACCESS Method made it easy for us to retrieve the needed information from the mainframe data server necessary for the migration work.

The resulting Excel workbooks proved to be a good tool for the users managing the migration work.

REFERENCES – AND RECOMMENDED READING

SAS Institute Inc., SAS® 9.3 Statements: Reference - FILENAME Statement, FTP Access Method
<http://support.sas.com/documentation/cdl/en/lestmtsref/63323/HTML/default/viewer.htm#p0v0ijxl1k6d4bn16cshtic7u4i3.htm>

SAS/ACCESS® 9.3 Interface to PC Files: Reference (SAS Institute Inc.), chapter on Libname Access and Excel Engines on Microsoft Windows.
<http://support.sas.com/documentation/cdl/en/acpcref/66980/HTML/default/viewer.htm#titlepage.htm>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Senior Specialist
Jesper Michelsen
Nykredit A/S
Kalvebid Brygge 47
1560 Copenhagen V
Work Phone: +45 4455 1362
Email: jmic@nykredit.dk

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.